/*
 *
 *
 *
 *
 *
 *
 */
#include <asm/boot.h>

.code16
.text
.global _start
_start:
	jmpl	$BOOTSEG, $start2

start2:
	movw	%cs, %ax
	movw 	%ax, %ds
	movw	%ax, %es
	movw	%ax, %ss
	movw	$0x7c00, %sp
	sti
	cld

	movw	$bugger_off_msg, %si

msg_loop:
	lodsb
	andb	%al, %al
	jz	die
	movb	$0xe, %ah
	movw	$7, %bx
	int	$0x10
	jmp	msg_loop

die:
	# Allow the user to press a key, then reboot
	xorw	%ax, %ax
	int	$0x16
	int	$0x19

	# int 0x19 should never return. In case it does anyway, 
	# invoke the BIOS reset code...
	ljmp	$0xf000, $0xfff0

bugger_off_msg:
	.ascii  "Direct booting from floppy is no longer supported. \r\n"
	.ascii	"Please use a boot loader program instead.\r\n"
	.ascii	"\n"
	.ascii	"Remove disk and press any key to reboot . . .\r\n"
	.byte	0

	# Kernel attributes; used by setup
	
	.org 497
setup_sects:	.byte	SETUPSECTS
root_flags:	.word	1
syssize:	.word	SYSSIZE
swap_dev:	.word	0
ram_size:	.word	0
vid_mode:	.word	0
root_dev:	.word	0
boot_flag:	.word	0xAA55


